Quartic Integral (1.3.11)

In[∘]:=

```
(1/Sqrt[2 π]) Integrate[Exp[-z^2/2 - g z^4/4], {z, -∞, ∞}]
```

Out[∘]=

$$\frac{e^{\frac{1}{8}/g}\,\text{BesselK}\left[\frac{1}{4}, \frac{1}{8\,g}\right]}{2\,\sqrt{g}\,\sqrt{\pi}} \quad \text{if } \text{Re}[g] > 0$$

Definitions

In[∘]:=

$$\text{quarticI}[g\_] := \frac{e^{\frac{1}{8}/g}\,\text{BesselK}\left[\frac{1}{4}, \frac{1}{8\,g}\right]}{2\,\sqrt{g}\,\sqrt{\pi}}$$

```
(*quarticIseries[g_, nmax_]:= Expand[(1/Sqrt[2π]) Distribute[Integrate[
    Normal[Series[Exp[-z^2/2 - g z^4/4], {g,0,nmax}]], {z,-∞,∞}]]]*)
```

$$\text{quarticIseries}[g\_, nmax\_] := \text{Sum}\left[\frac{(-1)^n\,\text{Gamma}\left[\frac{1}{2} + 2\,n\right]}{\sqrt{\pi}\,n!}\,g\text{^}n, \{n, 0, nmax\}\right]$$

In[∘]:=

```
boreltransf[series_, u_, nmax_] :=
 Module[{coefs, g}, coefs = CoefficientList[series[g, nmax], g];
  Sum[coefs[[n]] u^(n - 1) / Factorial[n - 1], {n, Length[coefs]}]]
borelpade[series_, u_, nmax_] := PadeApproximant[
  boreltransf[series, u, nmax], {u, 0, {Floor[nmax / 2], Floor[(nmax + 1) / 2]}}]
```

In[∘]:=

```
(* For Borel transform B[u] as a function of u only *)
zerosplot[borel_, color_ : Blue] :=
 Module[{zeros}, zeros = u /. NSolve[borel == 0, u];
  Print[zeros];
  ComplexListPlot[zeros, PlotStyle → Directive[color, PointSize[0.02]]]]
polesplot[borel_] := zerosplot[1 / borel, Red]
borelresum[borel_, g_?NumericQ] :=
 (1 / g) NIntegrate[Exp[-u / g] borel, {u, 0, ∞}]
```

Figure 3.1

In[•]:=
```
Table[{n, quarticI[g] - quarticIseries[g, n]}, {n, 8, 18}];
ListPlot[Abs[% /. g → 0.02]]
```

Out[•]=



Example 3.4

In[•]:=
```
(1 / Sqrt[2 π]) Integrate[
  Limit[D[Exp[-z^2 / 2 - g z^4 / 4], {g, n}] / Factorial[n], g → 0], {z, -∞, ∞}]
Sum[% u^n / Factorial[n], {n, 0, ∞}]
quarticI[0.2]
borelresum[%%, 0.2]
quarticIseries[g, 1] /. g → 0.2
```

Out[•]= $$\boxed{\frac{(-1)^n \,\mathrm{Gamma}\left[\frac{1}{2} + 2n\right]}{\sqrt{\pi}\; n!} \quad \text{if } \mathrm{Re}[n] > -\frac{1}{4}}$$

Out[•]= $$\frac{2\,\mathrm{EllipticK}\left[\frac{-1+\sqrt{1+4u}}{2\sqrt{1+4u}}\right]}{\pi\,(1+4u)^{1/4}}$$

Out[•]= 0.907985

Out[•]= 0.907985

Out[•]= 0.85

Table 3.1

In[•]:=
```
pade = borelpade[quarticIseries, u, 30];
Plot[pade, {u, -1, 1}]
 (*zeros*)
zerosplot[pade]
(*poles*)
polesplot[pade]
```

Out[•]=




```
{-60.7103, -9.33987, -3.63888, -1.94901, -1.23113, -0.861916, -0.648204, -0.514354,
 -0.425866, -0.365221, -0.322799, -0.293014, -0.272497, -0.259203, -0.251929}
```

Out[•]=




```
{-32.0488, -7.1462, -3.09185, -1.74073, -1.13247, -0.808605, -0.616808, -0.494764,
 -0.413172, -0.356821, -0.31722, -0.289375, -0.270249, -0.257987, -0.25149}
```

Out[•]=



Branch point at u = -1/4

In[•]:=
```
borelresum[pade, 0.2]
borelresum[pade, 0.4]
```

Out[•]= 0.907985

Out[•]= 0.857609

nmax = 10
0.9079854376030722`
0.8576207822712806`

nmax = 20
0.9079847775740948`
0.8576086008255066`

nmax = 30
0.9079847774309506`
0.8576085853876735`

Accuracy

quartic integral and approximations

For the quartic integral I(g), Eq. (1.3.11), we define a truncated series

$$\phi_n(g) = \sum_{k=0}^{n} a_k g^k.$$

Then, we can use the Borel-Pade resummation

$$s(\phi)_n(g) = g^{-1} \int e^{-u/g} P^{\phi}{}_n(u)\, du,$$

$$P^{\phi}{}_n(u) = [[n/2]/[(n+1)/2]]_{\hat{\phi}}(u),$$

and the Pade approximant for ϕ_n

$$R^{\phi}{}_n(g) = [[n/2]/[(n+1)/2]]_{\phi}(g).$$

In[•]:=
```
nmax = 10;
pade1 = borelpade[quarticIseries, u, nmax];
pade2 = PadeApproximant[quarticIseries[g, nmax],
    {g, 0, {Floor[nmax / 2], Floor[(nmax + 1) / 2]}}];
Plot[{quarticI[g], Evaluate[quarticIseries[g, nmax]], borelresum[pade1, g],
   pade2}, {g, 0, 1}, PlotRange → {Automatic, {0.75, 1}},
  PlotLegends → {"I(g)", "Truncated", "BorelPade", "Pade"}]
Plot[{quarticI[g], Evaluate[quarticIseries[g, nmax]], borelresum[pade1, g],
   pade2}, {g, 0, 20}, PlotRange → {Automatic, {0.3, 1}},
  PlotLegends → {"I(g)", "Truncated", "BorelPade", "Pade"}]
```

Out[•]=



Out[•]=



In[•]:=
```
(1 / Sqrt[2 π]) Integrate[Exp[- g z^4 / 4], {z, -∞, ∞}]
Series[quarticI[g], {g, ∞, 1}]
N[Series[pade2, {g, ∞, 1}]]
```

Out[•]= $\boxed{\dfrac{\text{Gamma}\left[\frac{1}{4}\right]}{2\, g^{1/4}\sqrt{\pi}} \quad \text{if } \text{Re}[g] > 0}$

Out[•]= $\dfrac{\text{Gamma}\left[\frac{1}{4}\right]\left(\frac{1}{g}\right)^{1/4}}{2\sqrt{\pi}} + \dfrac{\text{Gamma}\left[-\frac{1}{4}\right]\left(\frac{1}{g}\right)^{3/4}}{8\sqrt{\pi}} + O\left[\frac{1}{g}\right]^{5/4}$

Out[•]= $0.451709 + \dfrac{1.23804}{g} + O\left[\dfrac{1}{g}\right]^{2}$

$R^{\phi}{}_n$ and ϕ_n

$P^{\phi}{}_n$ and $\hat{\phi}$

Optimized perturbation theory

We consider the delta expansion (or optimized perturbation theory).

In[∘]:=
```
(1/Sqrt[2 π]) Integrate[Exp[-α z^2/2 + δ ((α - 1) z^2/2 - g z^4/4)], {z, -∞, ∞}]
```

Out[∘]=
$$\frac{e^{\frac{(\alpha+\delta-\alpha\,\delta)^2}{8\,g\,\delta}}\sqrt{\alpha+\delta-\alpha\,\delta}\;\text{BesselK}\left[\frac{1}{4},\frac{(\alpha+\delta-\alpha\,\delta)^2}{8\,g\,\delta}\right]}{2\sqrt{\pi}\,\sqrt{g\,\delta}}\quad \text{if}\;\; \text{Re}[g\,\delta]>0\;\&\&\;\text{Re}[\alpha+\delta-\alpha\,\delta]>0$$

The expansion is done with respect to δ and we will set δ=1 finally, as

$$\sum_{k=0}^{n} a_k(\alpha, g)\,\delta^k\,|_{\delta\to 1}.$$

We use the condition to fix the adjustable parameter α,

$$a_n(\alpha, g) = 0.$$

(fastest apparent convergence condition)

Numerical

Symbolic

In[∘]:=
```
N[Limit[g^(1/4) quarticI[g], g → ∞]]
```

Out[∘]= 1.02277

In[∘]:=
```
optcoef[n_, α_, g_] :=
  (1/Sqrt[2 π]) Integrate[Exp[-α z^2/2] ((α - 1) z^2/2 - g z^4/4)^n,
     {z, -∞, ∞}, Assumptions → g > 0] / Factorial[n]
```

In[•]:=
```
nmax = 3;
optcoef[nmax, α, g]
FullSimplify[Flatten[Solve[% == 0 && g > 0 && α > 0, α]]]
αsol = α /. %;
optg = FullSimplify[Sum[N[optcoef[n, αsol, g]], {n, 0, nmax - 1}]]

Plot[{quarticI[g], optg}, {g, 0, 100}, PlotLegends → {"I(g)", "OptimizedPT"}]
(* g→∞ *)
1.0227656721131686` - N[Limit[g^(1/4) optg, g → ∞]]

(* error *)
FullSimplify[optcoef[nmax + 1, αsol, g]]
Plot[%, {g, 0, 10}]
N[Limit[g^(1/4) %%, g → ∞]]
```

Out[•]= $\boxed{-\frac{5\left(693\,g^3-378\,g^2\,(-1+\alpha)\,\alpha+84\,g\,(-1+\alpha)^2\,\alpha^2-8\,(-1+\alpha)^3\,\alpha^3\right)}{128\,\alpha^{13/2}} \quad \text{if}\ \ \mathrm{Re}[\alpha]>0}$

Out[•]= $\left\{\alpha \to \boxed{\frac{1}{2}\left(1+\sqrt{1+g\ \boxed{16.6\ldots}}\right) \quad \text{if}\ \ g>0}\right\}$

Out[•]= $\boxed{\frac{11.3137+11.3137\,\sqrt{1.+16.5661\,g}+g\left(202.367+555.959\,g+108.655\,\sqrt{1.+16.5661\,g}\right)}{\left(1.+\sqrt{1.+16.5661\,g}\right)^{9/2}} \quad \text{if}\ \ g>0}$

Out[•]= 

Out[•]= 0.0186178

Out[•]= $\boxed{\frac{1575\,g^4\ \boxed{74.1\ldots}}{8\sqrt{2}\left(1+\sqrt{1+g\ \boxed{16.6\ldots}}\right)^{17/2}} \quad \text{if}\ \ g>0}$

Out[]=


Out[]= 0.0678503

Delta expansion as a special case of ODM

Order dependent mapping

Zinn-Justin notation

Marino notation

We now consider the order dependent mapping:

$$g = \rho \frac{\lambda}{(1-\lambda)^2},$$

where ρ is an adjustable parameter.
Then, from the original integral I(g), we have

In[]:=
```
(1/Sqrt[2 π]) (1 - λ)^(1/2)
 Integrate[Exp[-z^2/2 + λ (z^2/2 - ρ z^4/4)], {z, -∞, ∞}]
```

Out[]=
$$-\frac{e^{\frac{(-1+\lambda)^2}{8\lambda\rho}}\sqrt{1-\lambda}\,(-1+\lambda)\,\mathrm{BesselK}\left[\frac{1}{4},\frac{(-1+\lambda)^2}{8\lambda\rho}\right]}{\sqrt{2\pi}\,\sqrt{2-2\lambda}\,\sqrt{\lambda\rho}}\quad \text{if}\ \ \mathrm{Re}[\lambda]<1\ \&\&\ \mathrm{Re}[\lambda\rho]>0$$

The n-th approximant with $\rho = \rho_n$ is given by

$$\sum_{k=0}^{n} a_k(\rho_n)\,\lambda^k,\quad a_n(\rho_n) = 0.$$

In[]:=
```
N[Limit[g^(1/4) quarticI[g], g → ∞]]
```

Out[]= 1.02277

In[]:=
```
odmcoef[k_, ρ_] := If[k == 0, 1,
  1/(Sqrt[2 π] Factorial[k]) Integrate[Exp[-s^2/2] (s^2/2 - ρ s^4/4)^k, {s, -∞, ∞}]]
```

For odd nmax only, we may use
 (N)Solve[%==0, ρ, Reals].

For large nmax, (N)Solve for $\rho \in$ Complexes cannot find a “good” solution.
Then, the above assumption, $\rho \in$ Reals, is better.

```
In[•]:= (* For even nmax, ρ ∈ Complexes; Find a maximum value of Abs[ρ] *)
findmaxρ[ρlist_] :=
 Module[{ρabs}, ρabs = Table[Abs[ρ /. %[[i]]], {i, Length[ρlist]}];
  ρlist[[Position[ρabs, Max[ρabs]][[1, 1]]]]]
(* Solver *)
solveρ[coef_, symbolicQ_, complexQ_] := Module[{solve},
  If[symbolicQ, solve = Composition[FullSimplify, Solve], solve = NSolve];
  Flatten[If[complexQ,
    solve[% == 0 && Re[ρ] > 0 && Im[ρ] ≥ 0, ρ], solve[% == 0, ρ, Reals]]]]
```

In[•]:=
```
kval = 30;
symbolicQ = False; complexQ = True;
odmcoef[kval, ρ];
solveρ[%, symbolicQ, complexQ]
findmaxρ[%]
ρsol = ρ /. %;
N[ρsol kval]
N[1 / ρsol]
odm = Sum[odmcoef[k, ρsol] λ^k, {k, 0, kval}];
N[{odmcoef[kval, ρsol], odmcoef[kval + 1, ρsol]}]

Simplify[(1 - λ) ^ (1 / 2) odm /. {λ → (1 + 2 g / ρsol - Sqrt[1 + 4 g / ρsol]) / (2 g / ρsol)}];
Plot[{quarticI[g], Re[%], Im[%]}, {g, 0, 100},
 PlotLegends → {"I(g)", "ODM(re)", "ODM(im)"}]

(* g→∞ *)
1.0227656721131686` - (ρsol λ) ^ (1 / 4) odm /. λ → 1
```

Out[•]= {ρ → 0.00294738 + 0.0178833 i, ρ → 0.0057082 + 0.0184488 i, ρ → 0.00821163 + 0.0185224 i, ρ → 0.0105591 + 0.0182339 i, ρ → 0.0127693 + 0.0176401 i, ρ → 0.0148365 + 0.0167784 i, ρ → 0.016803 + 0.0156152 i, ρ → 0.0182587, ρ → 0.0185227, ρ → 0.0197114, ρ → 0.019993, ρ → 0.0207318, ρ → 0.0213616, ρ → 0.0213876, ρ → 0.0214366, ρ → 0.0219547, ρ → 0.0221063, ρ → 0.0229244, ρ → 0.0253938, ρ → 0.0254101, ρ → 0.0268104, ρ → 0.0269537, ρ → 0.027235}

Out[•]= ρ → 0.027235

Out[•]= 0.817049

Out[•]= 36.7175

Out[•]= $\{1.81197 \times 10^{-8}, -2.5253 \times 10^{-8}\}$

Out[•]=



Out[•]= -6.52711×10^{-9}